# Affine transformation of circular arc in 3D

Start with a quarter-circle of radius 1 centered at the origin and lying in the $$xz$$-plane:

 arc = ParametricPlot3D[{Cos[t], 0, Sin[t]}, {t, 0, π/2}]


I want to dilate this by a factor of 2 and shift the center to {3, 0, 0}, then show the result graphically.

The following does not work:

shiftAndDilate3D = AffineTransform[{2 IdentityMatrix[3], {3, 0, 0}}]

Graphics3D[GeometricTransformation[arc3D, shiftAndDilate3D]]


The error I get is that Graphics3DBox is not a Graphics3D primitive or directive.

What am I doing wrong?

• At least for this case, it is much better to just apply the affine transformation to the parametric equations directly: ParametricPlot3D[AffineTransform[{2 IdentityMatrix[3], {3, 0, 0}}] @ {Cos[t], 0, Sin[t]} // Evaluate, {t, 0, π/2}]. – J. M. is away Mar 8 at 1:46
• @J.M.iscomputer-less: That's a refreshingly different approach! I have to look at in the entire context of the more complicated thing I'm actually trying to do. It's worth making an answer! – murray Mar 8 at 1:55
• I think I kind of understand your confusion now, in light of this and your other question. One problem is that the docs do not give a complete and unambiguous list of primitives that one can point to and say: "these are primitives, and they are the only ones supported by GeometricTransformation[]"; this list mixes up directives and primitives, so that doesn't count. – J. M. is away Mar 8 at 23:56
• Possibly relevant: mathematica.stackexchange.com/questions/10957/… – Sjoerd C. de Vries Mar 9 at 14:07

arc = ParametricRegion[{Cos[t], 0, Sin[t]}, {{t, 0, \[Pi]/2}}];


The transformed arc:

shiftAndDilate3D = AffineTransform[{2 IdentityMatrix[3], {3, 0, 0}}];
new = TransformedRegion[arc, shiftAndDilate3D];


Visualization:

Show[
Region[arc, BaseStyle->Red],
Region[new, BaseStyle->Blue],
Axes->True
]


• How should one know from the Mathematica documentation that ParametricRegion is a "region" suitable for the 1st argument to TransformedRegion? (The documentation involving GeometricTransformatino, Region, etc., is sadly deficient. As with Image, these things seem to have been thrown into Mathematica without sufficient exposition, or perhaps even without sufficient coherence for the various kinds of things, on the one hand, and sufficient distinctions among them, on the other hand.) – murray Mar 7 at 20:22
• On the one hand, I like this answer better than the others because it is more direct. On the other hand, it uses yet a different type of object, namely, a "region". – murray Mar 7 at 20:23
• But I'm annoyed by the different method to style the resulting graphics display, namely, through use of the BaseStyle option of Region instead of the usual PlotStyle option of ParametricPlot3D and so many other graphics and graphics 3D functions. Moreover, why should it be necessary to wrap the result of a ParametricRegion expression with a Region? Either there's something fundamental here I don't understand, or else a whole slew of graphics-like or geometric-like constructs have been thrown into recent versions of Mathematica without sufficient rationalization of the whole domain. – murray Mar 7 at 20:36
• Further rant: Why does ParametricRegion take as 2nd argument a list of lists (of parameters and their extent), whereas ParametricPlot3D uses the parameter information as a list, then another list, etc.? – murray Mar 7 at 20:39
• @murray Lots of questions. For a brief overview of regions, see reference.wolfram.com/language/guide/GeometricComputation. Basically, a region is a computable object. Region is a function that displays the object (similar to RegionPlot). Show converts a region object to a Graphics/Graphics3D object. ParametricRegion has a different syntax because it supports arguments without bounds, e.g., ParametricRegion[{Cos[t], 0, Sin[t]}, t]. – Carl Woll Mar 7 at 20:48

You cannot apply those geometric transformations to the results of the plotting; instead, you should apply them to a Graphics primitive, e.g. the Line object generated by ParametricPlot, which you can extract using e.g. Cases:

arcLine =
First@Cases[ParametricPlot3D[{Cos[t], 0, Sin[t]}, {t, 0, π/2}], _Line, All]

Graphics3D[{
Red, arcLine,
Blue, GeometricTransformation[arcLine, shiftAndDilate3D]
}]


In red in the plot above is your original curve, in blue the transformed one.

• Where in the documentation does it explain what is, and what is not, a "geometric object"? – murray Mar 7 at 17:29
• @murray I was being loose with words there. I really meant a "Graphics primitive", such as those discussed in this guide: Graphics objects. I fixed it in the answer. – MarcoB Mar 7 at 18:20
• Alas, the doc page guide/GraphicsObjects does not fully identify or clarify what is, and what is not, a "graphics object". For example, given the answer by Carl Woll to my question, a ParametricRegion seems to qualify as such an object, yet is not listed on that guide page. So the documentation list of graphics objects doesn't specify what they are, and certainly -- and this is perhaps a real design defect -- the Head of these objects doesn't tell you that they are "graphics objects"! – murray Mar 7 at 19:53

Use arc3D[[1]] (which contains all the graphics primitives and their styles) as the first argument of GeometricTransformation:

arc3D = ParametricPlot3D[{Cos[t], 0, Sin[t]}, {t, 0, π/2}]
Show[arc3D,
Graphics3D[GeometricTransformation[arc3D[[1]], shiftAndDilate3D] /. l_Line :> {Orange, l}],
PlotRange -> All]


• I cannot reproduce that. Rather, I get an error message: "An improperly formatted option was encountered while reading a Graphics3DBox. The left-hand side of the option was not a symbol or string." – murray Mar 7 at 17:38
• @murray, forgot a comma before PlotRange (fixed now). – kglr Mar 7 at 17:41
• Sorry, I should have caught that! – murray Mar 7 at 19:47